

RECHERCHEH

Définition

La fonction **RECHERCHEH** permet de **rechercher** une **valeur** dans la **1^{ère} ligne** d'un **tableau** pour **renvoyer** une **donnée** qui se situe dans la **même colonne** que la **valeur recherchée**.

NB : **souvent imbriquée** dans la fonction **SI**, **SIERREUR** ou **SI.NON.DISP**, elle peut être **remplacée** par **RECHERCHEX** (ou **EQUIVX** imbriquée dans **INDEX**) qui est **plus complète** et **plus facile** à utiliser.

La fonction **RECHERCHEV** est **similaire** à **RECHERCHEH**, sauf qu'elle **recherche** une **valeur** dans la **1^{ère} colonne** d'un **tableau** pour **renvoyer** une **donnée** qui se situe dans la **même ligne**, alors que **RECHERCHEH** **recherche** une **valeur** dans la **1^{ère} ligne** d'un **tableau** pour **renvoyer** une **donnée** qui se situe dans la **même colonne**.

Syntaxe

RECHERCHEH(**Valeur_cherchée**;Table_matrice;No_index_lig;**[Valeur_proche]**)

Argument	Description
Valeur_cherchée (obligatoire)	Valeur à rechercher dans le tableau . Attention ! La valeur doit être située dans la 1^{ère} ligne du tableau , donc les données que tu veux extraire doivent être situées en bas de cette 1^{ère} ligne .
Table_matrice (obligatoire)	Tableau ou plage dans laquelle tu veux rechercher la valeur pour renvoyer une donnée qui se situe dans la même colonne .
No_index_lig (obligatoire)	N° de ligne (position) qui contient la donnée que tu veux renvoyer . Par exemple, la 1^{ère} ligne du tableau correspond à la valeur 1 .
[Valeur_proche] (facultatif)	Type de correspondance souhaité : ⇒ FAUX ou 0 (correspondance exacte) : cherche la 1^{ère} valeur exactement égale à la valeur . ⇒ VRAI , 1 ou omis (correspondance approximative) : cherche la plus grande valeur inférieure ou égale à la valeur . Attention ! Les valeurs de la 1^{ère} ligne du tableau doivent être triées par ordre croissant .

Exemple 1 : correspondance exacte (FAUX ou 0)

Rappel : **RECHERCHEH**(Valeur_cherchée;Table_matrice;No_index_lig;[Valeur_proche])

⇒ **FAUX** ou **0** (correspondance exacte) : cherche la 1^{ère} valeur exactement égale à la valeur.

Ce tableau liste les données de plusieurs salariés d'une entreprise. La valeur à rechercher est le matricule (A7 à A8) dans la 1^{ère} ligne (ligne 1) du tableau.

Tu veux renvoyer le prénom (cellule B7 à B8), le service (cellule C7 à C8), et le salaire brut mensuel (cellule D7 à D8) correspondant à chaque matricule du salarié (A7 à A8).

	A	B	C	D
1	Matricule	1001	1002	1003
2	Prénom	Emma	Guillaume	Julie
3	Service	Ressources Humaines	Comptabilité	Comptabilité
4	Salaire	1 900 €	2 200 €	2 500 €
5				
6	Matricule	Prénom	Service	Salaire
7	1003	(1) =RECHERCHEH(\$A7; \$B\$1:\$D\$4;2;FAUX)	=RECHERCHEH(\$A7; \$B\$1:\$D\$4;3;FAUX)	=RECHERCHEH(\$A7; \$B\$1:\$D\$4;4;FAUX)
8	1000	(2) #N/A =RECHERCHEH(\$A8; \$B\$1:\$D\$4;2;FAUX)	#N/A =RECHERCHEH(\$A8; \$B\$1:\$D\$4;3;FAUX)	#N/A =RECHERCHEH(\$A8; \$B\$1:\$D\$4;4;FAUX)

Remarques

⇒ (1) La valeur à rechercher ("1003" dans la cellule A7) est bien située dans la 1^{ère} ligne du tableau ("1003" dans la cellule D1). Les données que tu veux renvoyer sont donc en bas de cette 1^{ère} ligne.

⇒ (1) Le n° de ligne est égal à 2, car tu veux renvoyer le prénom du salarié ("Julie") qui est situé dans la 2^e ligne du tableau (ligne 2).

⇒ (1) Pourquoi les \$? Ils permettent de figer la colonne A de la valeur à rechercher et la plage du tableau pour étirer la formule jusqu'à D8 et modifier uniquement le n° de ligne (voir « Les références de cellule »).

⇒ (2) La valeur à rechercher ("1000") n'existe pas dans la 1^{ère} ligne du tableau ("Matricule"). Cela signifie que RECHERCHEH ne trouve pas ce qu'elle est censée rechercher, donc elle renvoie l'erreur #N/A. Tu peux alors utiliser la fonction SI.NON.DISP ou SIERREUR :

"=SI.NON.DISP(RECHERCHEH(\$A8;\$B\$1:\$D\$4;2;FAUX);"Mat. inconnu")" renvoie "Mat. inconnu" si l'erreur #N/A est renvoyée par la fonction RECHERCHEH.

"=SIERREUR(RECHERCHEH(\$A8;\$B\$1:\$D\$4;2;FAUX);"Mat. inconnu")" renvoie "Mat. inconnu" si n'importe quelle erreur est renvoyée par la fonction RECHERCHEH.

Exemple 2 : correspondance approximative (VRAI, 1 ou omis)

Rappel : **RECHERCHEH**(Valeur_cherchée;Table_matrice;No_index_lig;[Valeur_proche])

⇒ **VRAI, 1 ou omis** (correspondance approximative) : cherche la **plus grande valeur inférieure ou égale** à la **valeur**. **Attention** ! Les **valeurs** de la **1^{ère} ligne** du **tableau** doivent être **triés** par **ordre croissant**.

Ce tableau liste le **chiffre d'affaires réalisé (CA HT)** par une entreprise auprès de **plusieurs clients**. Les **valeurs à rechercher** sont les **CA HT (B2 à B5)** dans la **1^{ère} ligne (F1 à H1)** du **tableau**.

Tu veux **renvoyer** le **taux de remise applicable** auprès de **chaque client** (colonne **C**) en fonction du **CA HT** qu'il t'a apporté :

- ⇒ **De 0 € à 1 000 €** : **0%** de remise.
- ⇒ **De 1 000 € à 10 000 €** : **5%** de remise.
- ⇒ **10 000 € et plus** : **10%** de remise.

	A	B	C	D	E	F	G	H
1	Client	CA HT	Taux de remise		(3) CA HT	0 €	1 000 €	10 000 €
2	Amazon	350 €	(1) 0% (2) =RECHERCHEH(B2; F\$1:H\$2;2;VRAI)		Remise	0%	5%	10%
3	Apple	2 500 €	5%					
4	LVMH	13 000 €	10%					
5	Google	6 000 €	5% =RECHERCHEH(B5; F\$1:H\$2;2;VRAI)					

Remarques

- ⇒ (1) **RECHERCHEH** cherche la **plus grande valeur inférieure ou égale** à la **valeur** pour renvoyer la **donnée** : "0" est la **plus grande valeur** <= "350" donc **RECHERCHEH** renvoie "0%".
- ⇒ (2) Les **valeurs à rechercher** ("CA HT" de **B2 à B5**) sont bien situées dans la **1^{ère} ligne** du **tableau** ("CA HT" de **F1 à H1**). Les **données** que tu veux **renvoyer** sont donc **en bas** de cette **1^{ère} ligne**.
- ⇒ (2) Le **n° de ligne** est **égal à 2**, car tu veux **renvoyer** le **taux de remise** ("0%") qui est situé dans la **2^e ligne** du **tableau** (F2 à H2).
- ⇒ (2) **Pourquoi les \$?** Ils permettent de **figer** les **lignes 1 et 2** du **tableau** pour **étirer la formule** jusqu'à **C5** (voir « [Les références de cellule](#) »).
- ⇒ (3) Les **CA HT (F1 à H1)** sont bien **triés** par **ordre croissant**.